
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 9.4.5.4.7 which is part of the UTRAN test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_9_4_5_4_7
Test Group:
/12/UTRAN34_GMM_NIMTC.ttcn

ATS Version:
iwd-TTCN3-B2013-03_D15wk25 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
Intel 7360
Verification Status:
PASS
4 Corrections required for test case 9.4.5.4.7
4.1 Introduction

This section describes the changes required to make test case 9.4.5.4.7 run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D15wk25 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 f_TC_9_4_5_4_7_UTRAN()
	Record Name
	f_TC_9_4_5_4_7_UTRAN()

	Reason for change
	In the current TTCN implementation :
The Cell after creation is not being set to serving Cell power. This needs to be done.

The current TTCN prompts the user to insert a USIM for 34.121-1/9.4.5.4.7. This also needs to be changed to 34.123-1/9.4.5.4.7
The current TTCN calls function f_UTRAN34_AttachReject in preamble. This function performs an ATTACH REJ, but performs a LAU accept procedure and assigns a TMSI and a LAI to UE. As per the pre conditions, the UE should not have any LAI information. This needs to be corrected.

	Summary of change
	Called the function f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell)  to set the cell to serving cell power

Corrected the prompt for inserting a USIM for 34.123-1/9.4.5.4.7
Created a new function f_UTRAN34_AttachReject_LAU_REJ which performs a ATTACH REJ and a LAU REJ thereby clearing the USIM of any LAI.

	Source of change
	UTRAN34_LocationUpdatingNIMTC.ttcn

	MCC160 Comments
	1. Accepted

2. Accepted. Comment changed to "34.123-1 clause 9.4.5.4.7"
3. Accepted in principle. Used “f_UTRAN34_AttachReject(utran34_CellA, REJECT);” with updated function f_UTRAN34_AttachReject instead, see Additional change from MCC160


Before:
	function f_TC_9_4_5_4_7_UTRAN() runs on UTRAN_PTC
  { /* @sic Location updating / periodic search for HPLMN or higher priority PLMN / UE waits UE waits Minimum Periodic search timer */
    //PLMN / MCC / MNC contents
    var PLMN_Identity v_PLMN_A := {mcc := {0,0,1}, mnc := {0,1}};
    var PLMN_Identity v_PLMN_B := {mcc := {0,2,2}, mnc := {0,2}};
    var PLMN_Identity v_PLMN_C := {mcc := {0,0,1}, mnc := {1,0}};
    var PLMN_Identity v_PLMN_D := {mcc := {0,0,1}, mnc := {1,1}};
    var PLMN_Identity v_PLMN_E := {mcc := {0,0,1}, mnc := {3,0}};
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    // Set NMO II in all cells
    f_UTRAN_CellInfo_SetNMO(utran34_CellA, tsc_NMO_II);
    f_UTRAN_CellInfo_SetNMO(utran34_CellB, tsc_NMO_II);
    f_UTRAN_CellInfo_SetNMO(utran34_CellC, tsc_NMO_II);
    f_UTRAN_CellInfo_SetNMO(utran34_CellD, tsc_NMO_II);
    f_UTRAN_CellInfo_SetNMO(utran34_CellE, tsc_NMO_II);
    //Set PLMN in all cells
    f_UTRAN_CellInfo_SetPLMNIdentity(utran34_CellA, v_PLMN_A);
    f_UTRAN_CellInfo_SetPLMNIdentity(utran34_CellB, v_PLMN_B);
    f_UTRAN_CellInfo_SetPLMNIdentity(utran34_CellC, v_PLMN_C);
    f_UTRAN_CellInfo_SetPLMNIdentity(utran34_CellE, v_PLMN_E);
    f_UTRAN_CellInfo_SetPLMNIdentity(utran34_CellD, v_PLMN_D);
    //Initialised test channels
    f_UTRAN34_SetSpecificFrequency(utran34_CellA, 1);
    f_UTRAN34_SetSpecificFrequency(utran34_CellB, 2);
    f_UTRAN34_SetSpecificFrequency(utran34_CellC, 3);
    f_UTRAN34_SetSpecificFrequency(utran34_CellD, 4);
    //initially Cells A, B and C shall not be broadcasting
    // CellA started
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    // CellB started
    f_UTRAN34_SS_CreateCellDCH(utran34_CellB);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellB);
    // CellC started
    f_UTRAN34_SS_CreateCellDCH(utran34_CellC);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellC);
    // CellD started
    f_UTRAN34_SS_CreateCellDCH(utran34_CellD);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellD);
    // CellE is not activated
    //Equip UE with a USIM containing default values except for those shown in clause 9.4.3.7.4
    f_UT_USIM_Insert(UT, "34.121-1 clause 9.4.5.4.7.4");
    // The location area information on the USIM is "deleted"
    f_UTRAN34_AttachReject (utran34_CellA);

    //the UE is deactivated.
    f_UT_SwitchOffUE(UT, false);    
    f_UTRAN_TestBody_Set(true);
    fl_TC_9_4_5_4_7_Body();
    f_UTRAN_TestBody_Set(false);
    // Release RRC connection on cell 1 and release the cells
    f_UTRAN34_Postamble(utran34_CellA, U1_IDLE);
    f_UTRAN_ReleaseCell(utran34_CellB, f_UTRAN_CellInfo_GetConfigType(utran34_CellB));
    f_UTRAN_ReleaseCell(utran34_CellC, f_UTRAN_CellInfo_GetConfigType(utran34_CellC));
    f_UTRAN_ReleaseCell(utran34_CellD, f_UTRAN_CellInfo_GetConfigType(utran34_CellD));
  } // f_TC_9_4_5_4_7_UTRAN


After:

	function f_TC_9_4_5_4_7_UTRAN() runs on UTRAN_PTC
  { /* @sic Location updating / periodic search for HPLMN or higher priority PLMN / UE waits UE waits Minimum Periodic search timer */
    //PLMN / MCC / MNC contents
    var PLMN_Identity v_PLMN_A := {mcc := {0,0,1}, mnc := {0,1}};
    var PLMN_Identity v_PLMN_B := {mcc := {0,2,2}, mnc := {0,2}};
    var PLMN_Identity v_PLMN_C := {mcc := {0,0,1}, mnc := {1,0}};
    var PLMN_Identity v_PLMN_D := {mcc := {0,0,1}, mnc := {1,1}};
    var PLMN_Identity v_PLMN_E := {mcc := {0,0,1}, mnc := {3,0}};
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    // Set NMO II in all cells
    f_UTRAN_CellInfo_SetNMO(utran34_CellA, tsc_NMO_II);
    f_UTRAN_CellInfo_SetNMO(utran34_CellB, tsc_NMO_II);
    f_UTRAN_CellInfo_SetNMO(utran34_CellC, tsc_NMO_II);
    f_UTRAN_CellInfo_SetNMO(utran34_CellD, tsc_NMO_II);
    f_UTRAN_CellInfo_SetNMO(utran34_CellE, tsc_NMO_II);
    //Set PLMN in all cells
    f_UTRAN_CellInfo_SetPLMNIdentity(utran34_CellA, v_PLMN_A);
    f_UTRAN_CellInfo_SetPLMNIdentity(utran34_CellB, v_PLMN_B);
    f_UTRAN_CellInfo_SetPLMNIdentity(utran34_CellC, v_PLMN_C);
    f_UTRAN_CellInfo_SetPLMNIdentity(utran34_CellE, v_PLMN_E);
    f_UTRAN_CellInfo_SetPLMNIdentity(utran34_CellD, v_PLMN_D);
    //Initialised test channels
    f_UTRAN34_SetSpecificFrequency(utran34_CellA, 1);
    f_UTRAN34_SetSpecificFrequency(utran34_CellB, 2);
    f_UTRAN34_SetSpecificFrequency(utran34_CellC, 3);
    f_UTRAN34_SetSpecificFrequency(utran34_CellD, 4);
    //initially Cells A, B and C shall not be broadcasting
    // CellA started
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell);//WA#9_4_5_4_7
    // CellB started
    f_UTRAN34_SS_CreateCellDCH(utran34_CellB);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellB);
    // CellC started
    f_UTRAN34_SS_CreateCellDCH(utran34_CellC);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellC);
    // CellD started
    f_UTRAN34_SS_CreateCellDCH(utran34_CellD);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellD);
    // CellE is not activated
    //Equip UE with a USIM containing default values except for those shown in clause 9.4.3.7.4
    f_UT_USIM_Insert(UT, "34.123-1 clause 9.4.5.4.7.4");//WA#9_4_5_4_7 Replaced "34.123-1 clause 9.4.3.7.4"
    // The location area information on the USIM is "deleted"
    //WA#9_4_5_4_7 f_UTRAN34_AttachReject (utran34_CellA);
    f_UTRAN34_AttachReject_LAU_REJ (utran34_CellA); //WA#9_4_5_4_7
    //the UE is deactivated.
    f_UT_SwitchOffUE(UT, false);    
    f_UTRAN_TestBody_Set(true);
    fl_TC_9_4_5_4_7_Body();
    f_UTRAN_TestBody_Set(false);
    // Release RRC connection on cell 1 and release the cells
    f_UTRAN34_Postamble(utran34_CellA, U1_IDLE);
    f_UTRAN_ReleaseCell(utran34_CellB, f_UTRAN_CellInfo_GetConfigType(utran34_CellB));
    f_UTRAN_ReleaseCell(utran34_CellC, f_UTRAN_CellInfo_GetConfigType(utran34_CellC));
    f_UTRAN_ReleaseCell(utran34_CellD, f_UTRAN_CellInfo_GetConfigType(utran34_CellD));
  } // f_TC_9_4_5_4_7_UTRAN


4.3 fl_TC_9_4_5_4_7_Body()
	Record Name
	fl_TC_9_4_5_4_7_Body()

	Reason for change
	In the current TTCN implementaion, the LAU REQ in Step 4 expects a LAI and a TMSI. This is now incorrect because of Change 4.2 ,Summary of Change Point 3. Since the UE has received a LAU REJ and a ATTACH REJ in preamble, the UE wont have a TMSI or a LAI. This needs to be corrected.

After the preamble the UE is switched off and the Cell A is set to non-suitable cell power and Cell D to suitable Cell power. The Cell D needs to be reconfigured before the UE is switched on.

In current TTCN implementation for Step 2 of expected test sequence, after the UE is switched ON, a timer of 8 minutes is started. This is not required. The Prose doesn’t mention that the UE must respond within 8 minutes. This needs to be removed.

In the TTCN implementation of the LAU procedure and ATTACH procedure (Step 8), the cell ID is being sent as utran_CellDedicated. This needs to be changed to utran_CellD. 

After Step 13 the Cell A is set to suitable neighbouring cell power and the UE is expected to camp on it within 8 minutes.  The Cell D needs to be reconfigured and a timer check needs to be implemented to check that the UE camps within 8 minutes. This needs to be implemented.
In the TTCN implementation of the LAU ACCEPT at Step 17, the cell ID is being sent as utran_CellDedicated. This needs to be changed to utran_CellA

	Summary of change
	Replaced v_LAI_D and cr_MobileIDTmsi_lv(px_TMSI_Def) with * respectively in Step 4.
Called the function f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellD, utran34_CellA) before the UE is sswitched ON after preamble.

Disabled the starting and stopping of timer t_Wait8minutes(marked in green)

Replaced the Cell ID in Step 8 with utran34_CellD instead of utran_CellDedicated.

Called the function f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellA, utran34_CellD) before Step 13. Also implemented a check to ensure that the UE camps within 8 minutes. Disabled the call to function f_UTRAN34_RRC_ConnEst(utran34_CellA, registration), (marked in green)

Replaced the Cell ID in Step 17 with utran34_CellA instead of utran_CellDedicated.

	Source of change
	UTRAN34_GMM_NIMTC.ttcn

	MCC160 Comments
	1. Accepted in principle. Used the predefined parameters of the template instead of ‘*’ which could lead on a runtime error
2. Accepted

3. Accepted in principle. Delay implemented right after step2 as per prose.
4. Accepted

5. Accepted in principle. Step 14 is rewritten but the timer of 8 minutes needs to be active until Step15 is done. See implementation below. Added enhancements on the interleave sections.
6. Accepted


Before:
	function fl_TC_9_4_5_4_7_Body() runs on UTRAN_PTC
  {
    var RRC_DATA_IND v_RRC_DataInd_RAU;
    var RRC_DATA_IND v_RRC_DataInd_CS;
    var RRC_DATA_IND v_RRC_DataInd_PS;
    var NAS_PlmnId v_PLMN_IdentityA;
    var NAS_PlmnId v_PLMN_IdentityD;
    var NAS_PlmnId v_PLMN_IdentityE;
    var B3_Type v_UpdateType_Ra := '000'B;
    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step4;
    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step15;
    var template (present) LocAreaId v_LAI_A;
    var template (present) LocAreaId v_LAI_D;
    var boolean v_GMM_Attach_received := false;
    var boolean v_RAU_received := false;
    var boolean v_LAU_received := false;
    var float v_Wait8minutes := 480.0;
    timer t_WaitForGMM := 5.0;
    timer t_Wait8minutes;
    // Variables to be used with cell A
    v_PLMN_IdentityA := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA));
    v_LAI_A := cr_LAI(omit, v_PLMN_IdentityA, f_UTRAN_CellInfo_GetLAC(utran34_CellA));
    // Variables to be used with cell D
    v_PLMN_IdentityD := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellD));
    v_LAI_D := cr_LAI(omit, v_PLMN_IdentityD, f_UTRAN_CellInfo_GetLAC(utran34_CellD));
    //Used for Equivalent PLMN List
    v_PLMN_IdentityE := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellE));
    // TMSI used throughout the test case is px_TMSI_Def;
    // new TMSI allocated will therefore always be the same
    v_U_LocationUpdatingRequest_Step4 :=  cr_U_LocationUpdatingRequestAll(?, // CKSN = initial value
                                                                          cr_LocUpdTypeNormal,
                                                                          v_LAI_D, // LAI = d 
                                                                          cr_MS_Clsmk1_Def,
                                                                          cr_MobileIdTMSI_lv(px_TMSI_Def), // mobile identity = TMSI
                                                                          cr_MS_Clsmk2_Any_tlv('33'O) ifpresent);
    v_U_LocationUpdatingRequest_Step15 :=  cr_U_LocationUpdatingRequestAll(?, // CKSN = initial value
                                                                           cr_LocUpdTypeNormal,
                                                                           v_LAI_D, // LAI = d 
                                                                           cr_MS_Clsmk1_Def,
                                                                           cr_MobileIdTMSI_lv(px_TMSI_Def),// mobile identity = TMSI
                                                                           cr_MS_Clsmk2_Any_tlv('33'O) ifpresent);
    //@siclog "Step 1" siclog@
    //Set the cell type of cell B to the "Serving cell". Set the cell type of cell A to the "non-suitable cell".
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationNonSuitableCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationNonSuitableCell);
    f_UTRAN_SetCellPower(utran34_CellC, tsc_AttenuationNonSuitableCell);
    f_UTRAN_SetCellPower(utran34_CellD, tsc_AttenuationSuitableNeighbourCell);    
    //@siclog "Step 2" siclog@
    //The UE is switched on by either using the Power Switch or by applying power.
    //If PS Mode: The subsequent GMM attach should be accepted with "Equivalent PLMNs": PLMN E.
    //If PS mode: a routing area updating procedure should be performed.
    f_UT_SwitchOnUE(UT, false);
    //The SS shall wait for 8 minutes during which no messages should be received
    t_Wait8minutes.start(v_Wait8minutes);

    //@siclog "Step 3" siclog@
    //The SS verifies that the IE "Establishment cause" in the received RRC CONNECTION REQUEST message is set to "Registration"
    f_UTRAN34_RRC_ConnEst(utran34_CellD, registration);
    //If PS Mode: The subsequent GMM attach should be accepted with "Equivalent PLMNs": PLMN E.
    if (pc_PS == true) {
      t_WaitForGMM.start;
      interleave { //Receive in any order Location Area Update or GMM Attach
        //GMM attach
        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt,
                                                                                                                 tsc_I_AttachTypeCombined)),
                                                                                               cr_MobileIdAny(omit),
                                                                                               cr_RAI(omit),
                                                                                               ?))) -> value v_RRC_DataInd_PS
          {
            v_GMM_Attach_received := true;
          }
        //Location Area Update
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step4)) -> value v_RRC_DataInd_CS
          {
            v_LAU_received := true;
          }
      }      
    }
    else { // only CS
      //@siclog "Step 4 -8" siclog@
      //LOCATION UPDATING. "Location Update Type": Normal
      //The SS starts integrity protection. "Equivalent PLMNs": PLMN E
      f_UTRAN34_LocationUpdate(utran_CellDedicated, true, TMSI, v_U_LocationUpdatingRequest_Step4, cs_PLMN_List_1PLMN(v_PLMN_IdentityE));
    }
    if (v_GMM_Attach_received == true) {//GMM attach should be accepted with "Equivalent PLMNs": PLMN E.
      f_UTRAN_GMM_Authentication();
      f_UTRAN_RRC_Security(utran34_CellD,                        // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,
                           ps_domain);      
      f_UTRAN_AttachAccept(utran_CellDedicated, v_RRC_DataInd_PS.msg.attachRequest.attachType.attachType,
                           cs_PLMN_List_1PLMN(v_PLMN_IdentityE));       // ATTACH ACCEPT, ATTACH COMPLETE
      t_WaitForGMM.stop;
      v_GMM_Attach_received := false;
    }
    if (v_LAU_received == true) {
      //Send Location Update Accept      
      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran_CellDedicated, v_RRC_DataInd_CS.ttcn_start, true, TMSI, cs_PLMN_List_1PLMN(v_PLMN_IdentityE));
      v_LAU_received := false;
    }
    //@siclog "Step 9" siclog@
    //The SS releases the RRC connection
    f_UTRAN34_RRC_ConnRel(utran34_CellD, cell_Dch);
    f_UTRAN_CheckNoDataInd(v_Wait8minutes, "Test Case 9.4.5.4.7, Step 10");
    //@siclog "Step 10" siclog@
    //The SS waits a period of 8 minutes after the UE is switched on, this allowing the UE to make its first periodic search
    t_Wait8minutes.timeout;

    //@siclog "Step 11" siclog@
    //Set the cell type of cell B to the "Suitable neighbour cell"
    //Set the cell type of cell C to the "Suitable neighbour cell"
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationSuitableNeighbourCell);
    f_UTRAN_SetCellPower(utran34_CellC, tsc_AttenuationSuitableNeighbourCell);
    //@siclog "Step 12" siclog@    
    //The SS shall wait for 8 minutes during which no messages should be received
    f_UTRAN_CheckNoDataInd(v_Wait8minutes, "Test Case 9.4.5.4.7, Step 12");
    //@siclog "Step 13" siclog@
    //Set the cell type of cell A to the "Suitable neighbour cell"
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationSuitableNeighbourCell);
    //Within 8 minutes after step 13, the following messages shall be sent and received on Cell A
    //The lower boundary of time T shall not be checked in this test step
    t_Wait8minutes.start(v_Wait8minutes);
    //@siclog "Step 14" siclog@
    //The SS verifies that the IE "Establishment cause" in the received RRC CONNECTION REQUEST message is set to "Registration".
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);

    if (pc_PS == true) {
      t_WaitForGMM.start;
      // wait for RAU and LU in any order
      interleave {
        //Receive Routing Area Update Request message
        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                                  tsc_RB3,
                                                  cr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, v_UpdateType_Ra),
                                                                    cr_RAI(omit),
                                                                    cr_PTMSI_SignatureAny ifpresent,
                                                                    ?))) -> value v_RRC_DataInd_RAU
          {
            v_RAU_received := true;
          }
        //Wait for Location Updating Request message
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step15)) -> value v_RRC_DataInd_CS
          {
            v_LAU_received := true;
          }
      }
    }
    else { // only CS so just wait for LU
      //@siclog "Step 15-18" siclog@
      //LOCATION UPDATING REQUEST
      f_UTRAN34_LocationUpdate(utran_CellDedicated, true, TMSI, v_U_LocationUpdatingRequest_Step15);
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.4.7, Step 15");
    }
    if (v_RAU_received == true) {
      t_WaitForGMM.stop;
      //SECURITY MODE COMMAND and SECURITY MODE COMPLETE
      f_UTRAN_GMM_Authentication();
      f_UTRAN_RRC_Security(utran34_CellA, true, ps_domain);
      //Send ROUTING AREA UPDATE ACCEPT
      U_Dc.send(cas_DataReq_PS(utran_CellDedicated, tsc_RB3, cs_U_RA_UpdAcc3(cs_GMM_UpdateResult(v_UpdateType_Ra),
                                                                             cs_RAI(omit,
                                                                                    cs_LAI(omit, v_PLMN_IdentityA, f_UTRAN_CellInfo_GetLAC(utran34_CellA)),
                                                                                    f_UTRAN_CellInfo_GetRAC(utran34_CellA)),
                                                                             cs_PTMSI_Signature(px_PTMSI_SigDef),
                                                                             cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def),
                                                                             v_RRC_DataInd_RAU.msg.routingAreaUpdateRequest.pDP_ContextStatus)));
      //Receive ROUTING AREA UPDATE COMPLETE
      U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));
    }
    if (v_LAU_received == true) {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.4.7, Step 15");
      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran_CellDedicated, v_RRC_DataInd_CS.ttcn_start, true, TMSI);
    }
    //@siclog "Step 19" siclog@
    //The SS releases the RRC connection
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    t_Wait8minutes.stop;
  } // fl_TC_9_4_5_4_7_Body


After:

	function fl_TC_9_4_5_4_7_Body() runs on UTRAN_PTC
  {
    var RRC_DATA_IND v_RRC_DataInd_RAU;
    var RRC_DATA_IND v_RRC_DataInd_CS;
    var RRC_DATA_IND v_RRC_DataInd_PS;
    var NAS_PlmnId v_PLMN_IdentityA;
    var NAS_PlmnId v_PLMN_IdentityD;
    var NAS_PlmnId v_PLMN_IdentityE;
    var B3_Type v_UpdateType_Ra := '000'B;
    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step4;
    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step15;
    var template (present) LocAreaId v_LAI_A;
    var template (present) LocAreaId v_LAI_D;
    var boolean v_GMM_Attach_received := false;
    var boolean v_RAU_received := false;
    var boolean v_LAU_received := false;
    var float v_Wait8minutes := 480.0;
    timer t_WaitForGMM := 5.0;
    timer t_Wait8minutes;
    var InitialUE_Identity v_InitialUE_Id; //WA#9_4_5_4_7
    var U_RLC_TM_IND v_RLC_TR_DATA_IND;//WA#9_4_5_4_7
    // Variables to be used with cell A
    v_PLMN_IdentityA := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA));
    v_LAI_A := cr_LAI(omit, v_PLMN_IdentityA, f_UTRAN_CellInfo_GetLAC(utran34_CellA));
    // Variables to be used with cell D
    v_PLMN_IdentityD := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellD));
    v_LAI_D := cr_LAI(omit, v_PLMN_IdentityD, f_UTRAN_CellInfo_GetLAC(utran34_CellD));
    //Used for Equivalent PLMN List
    v_PLMN_IdentityE := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellE));
    // TMSI used throughout the test case is px_TMSI_Def;
    // new TMSI allocated will therefore always be the same
    v_U_LocationUpdatingRequest_Step4 :=  cr_U_LocationUpdatingRequestAll(?, // CKSN = initial value
                                                                          cr_LocUpdTypeNormal,
                                                                          *, // LAI = d WA#9_4_5_4_7 v_LAI_D
                                                                          cr_MS_Clsmk1_Def,
                                                                          *, //WA#9_4_5_4_7cr_MobileIdTMSI_lv(px_TMSI_Def),// mobile identity = TMSI
                                                                          cr_MS_Clsmk2_Any_tlv('33'O) ifpresent);
    v_U_LocationUpdatingRequest_Step15 :=  cr_U_LocationUpdatingRequestAll(?, // CKSN = initial value
                                                                           cr_LocUpdTypeNormal,
                                                                           v_LAI_D, // LAI = d 
                                                                           cr_MS_Clsmk1_Def,
                                                                           cr_MobileIdTMSI_lv(px_TMSI_Def),// mobile identity = TMSI
                                                                           cr_MS_Clsmk2_Any_tlv('33'O) ifpresent);
    //@siclog "Step 1" siclog@
    //Set the cell type of cell B to the "Serving cell". Set the cell type of cell A to the "non-suitable cell".
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellD, utran34_CellA);//WA#9_4_5_4_7
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationNonSuitableCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationNonSuitableCell);
    f_UTRAN_SetCellPower(utran34_CellC, tsc_AttenuationNonSuitableCell);
    f_UTRAN_SetCellPower(utran34_CellD, tsc_AttenuationSuitableNeighbourCell);    
    //@siclog "Step 2" siclog@
    //The UE is switched on by either using the Power Switch or by applying power.
    //If PS Mode: The subsequent GMM attach should be accepted with "Equivalent PLMNs": PLMN E.
    //If PS mode: a routing area updating procedure should be performed.
    f_UT_SwitchOnUE(UT, false);
    //The SS shall wait for 8 minutes during which no messages should be received
    //WA#9_4_5_4_7 t_Wait8minutes.start(v_Wait8minutes);
    //@siclog "Step 3" siclog@
    //The SS verifies that the IE "Establishment cause" in the received RRC CONNECTION REQUEST message is set to "Registration"
    f_UTRAN34_RRC_ConnEst(utran34_CellD, registration);
    //If PS Mode: The subsequent GMM attach should be accepted with "Equivalent PLMNs": PLMN E.
    if (pc_PS == true) {
      t_WaitForGMM.start;
      interleave { //Receive in any order Location Area Update or GMM Attach
        //GMM attach
        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt,
                                                                                                                 tsc_I_AttachTypeCombined)),
                                                                                               cr_MobileIdAny(omit),
                                                                                               cr_RAI(omit),
                                                                                               ?))) -> value v_RRC_DataInd_PS
          {
            v_GMM_Attach_received := true;
          }
        //Location Area Update
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step4)) -> value v_RRC_DataInd_CS
          {
            v_LAU_received := true;
          }
      }      
    }
    else { // only CS
      //@siclog "Step 4 -8" siclog@
      //LOCATION UPDATING. "Location Update Type": Normal
      //The SS starts integrity protection. "Equivalent PLMNs": PLMN E
      f_UTRAN34_LocationUpdate(utran_CellDedicated, true, TMSI, v_U_LocationUpdatingRequest_Step4, cs_PLMN_List_1PLMN(v_PLMN_IdentityE));
    }
    if (v_GMM_Attach_received == true) {//GMM attach should be accepted with "Equivalent PLMNs": PLMN E.
      f_UTRAN_GMM_Authentication();
      f_UTRAN_RRC_Security(utran34_CellD,                        // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,
                           ps_domain);      
      f_UTRAN_AttachAccept(/*utran_CellDedicated WA#9_4_5_4_7*/ utran34_CellD, v_RRC_DataInd_PS.msg.attachRequest.attachType.attachType,
                           cs_PLMN_List_1PLMN(v_PLMN_IdentityE));       // ATTACH ACCEPT, ATTACH COMPLETE
      t_WaitForGMM.stop;
      v_GMM_Attach_received := false;
    }
    if (v_LAU_received == true) {
      //Send Location Update Accept      
      f_UTRAN34_LocationUpdate_WithoutLAUReq(/*utran_CellDedicated WA#9_4_5_4_7*/ utran34_CellD, v_RRC_DataInd_CS.ttcn_start, true, TMSI, cs_PLMN_List_1PLMN(v_PLMN_IdentityE));
      v_LAU_received := false;
    }
    //@siclog "Step 9" siclog@
    //The SS releases the RRC connection
    f_UTRAN34_RRC_ConnRel(utran34_CellD, cell_Dch);
    f_UTRAN_CheckNoDataInd(v_Wait8minutes, "Test Case 9.4.5.4.7, Step 10");
    //@siclog "Step 10" siclog@
    //The SS waits a period of 8 minutes after the UE is switched on, this allowing the UE to make its first periodic search
    //WA#9_4_5_4_7 t_Wait8minutes.timeout;
    //@siclog "Step 11" siclog@
    //Set the cell type of cell B to the "Suitable neighbour cell"
    //Set the cell type of cell C to the "Suitable neighbour cell"
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationSuitableNeighbourCell);
    f_UTRAN_SetCellPower(utran34_CellC, tsc_AttenuationSuitableNeighbourCell);
    //@siclog "Step 12" siclog@    
    //The SS shall wait for 8 minutes during which no messages should be received
    f_UTRAN_CheckNoDataInd(v_Wait8minutes, "Test Case 9.4.5.4.7, Step 12");
    //@siclog "Step 13" siclog@
    //Set the cell type of cell A to the "Suitable neighbour cell"
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellA, utran34_CellD);//WA#9_4_5_4_7
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationSuitableNeighbourCell);
    //Within 8 minutes after step 13, the following messages shall be sent and received on Cell A
    //The lower boundary of time T shall not be checked in this test step
    t_Wait8minutes.start(v_Wait8minutes);
    //WA#9_4_5_4_7 : Change start
    alt{ 
        [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                      tsc_RB0,
                                      cr_108_RRC_ConnReq_r8(registration))) -> value v_RLC_TR_DATA_IND
        {
          v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
          f_UTRAN_CheckReleaseIndicator(v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator);
          f_UTRAN_MobileInfo_UE_Release_Set(v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator);
          f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);
        }        
        [] t_Wait8minutes.timeout
          {
            f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 9.4.5.4.7 : The UE did not initiate a rrcConnection within 8 minutes, Step 14");
          }
    }
    //WA#9_4_5_4_7 : Change Stop
    //@siclog "Step 14" siclog@
    //The SS verifies that the IE "Establishment cause" in the received RRC CONNECTION REQUEST message is set to "Registration".
    //WA#9_4_5_4_7 f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    if (pc_PS == true) {
      t_WaitForGMM.start;
      // wait for RAU and LU in any order
      interleave {
        //Receive Routing Area Update Request message
        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                                  tsc_RB3,
                                                  cr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, v_UpdateType_Ra),
                                                                    cr_RAI(omit),
                                                                    cr_PTMSI_SignatureAny ifpresent,
                                                                    ?))) -> value v_RRC_DataInd_RAU
          {
            v_RAU_received := true;
          }
        //Wait for Location Updating Request message
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step15)) -> value v_RRC_DataInd_CS
          {
            v_LAU_received := true;
          }
      }
    }
    else { // only CS so just wait for LU
      //@siclog "Step 15-18" siclog@
      //LOCATION UPDATING REQUEST
      f_UTRAN34_LocationUpdate(utran_CellDedicated, true, TMSI, v_U_LocationUpdatingRequest_Step15);
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.4.7, Step 15");
    }
    if (v_RAU_received == true) {
      t_WaitForGMM.stop;
      //SECURITY MODE COMMAND and SECURITY MODE COMPLETE
      f_UTRAN_GMM_Authentication();
      f_UTRAN_RRC_Security(utran34_CellA, true, ps_domain);
      //Send ROUTING AREA UPDATE ACCEPT
      U_Dc.send(cas_DataReq_PS(utran_CellDedicated, tsc_RB3, cs_U_RA_UpdAcc3(cs_GMM_UpdateResult(v_UpdateType_Ra),
                                                                             cs_RAI(omit,
                                                                                    cs_LAI(omit, v_PLMN_IdentityA, f_UTRAN_CellInfo_GetLAC(utran34_CellA)),
                                                                                    f_UTRAN_CellInfo_GetRAC(utran34_CellA)),
                                                                             cs_PTMSI_Signature(px_PTMSI_SigDef),
                                                                             cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def),
                                                                             v_RRC_DataInd_RAU.msg.routingAreaUpdateRequest.pDP_ContextStatus)));
      //Receive ROUTING AREA UPDATE COMPLETE
      U_Dc.receive(car_PS_UplinkDirectTransfer/*WA#9_4_5_4_7 car_PS_InitDirectTransfer*/(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));
    }
    if (v_LAU_received == true) {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.4.7, Step 15");
      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran34_CellA /*WA#9_4_5_4_7 utran_CellDedicated*/, v_RRC_DataInd_CS.ttcn_start, true, TMSI);
    }
    //@siclog "Step 19" siclog@
    //The SS releases the RRC connection
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    t_Wait8minutes.stop;
  } // fl_TC_9_4_5_4_7_Body


MCC160 implementation
4.3 Change 5
	  function fl_TC_9_4_5_4_7_Body() runs on UTRAN_PTC

  {

    var RRC_DATA_IND v_RRC_DataInd_RAU;

    var RRC_DATA_IND v_RRC_DataInd_CS;

    var RRC_DATA_IND v_RRC_DataInd_PS;

    var NAS_PlmnId v_PLMN_IdentityA;

    var NAS_PlmnId v_PLMN_IdentityD;

    var NAS_PlmnId v_PLMN_IdentityE;

    var B3_Type v_UpdateType_Ra := '000'B;

    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step4;

    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step15;

    var template (present) LocAreaId v_LAI_A;

    var template (present) LocAreaId v_LAI_D;

    var InitialUE_Identity v_InitialUE_Id;

    var U_RLC_TM_IND v_RLC_TR_DATA_IND;

    var float v_Wait8minutes := 480.0;

    timer t_WaitForGMM := 5.0;

    timer t_Wait8minutes;

    // Variables to be used with cell A

    v_PLMN_IdentityA := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA));

    v_LAI_A := cr_LAI(omit, v_PLMN_IdentityA, f_UTRAN_CellInfo_GetLAC(utran34_CellA));

    // Variables to be used with cell D

    v_PLMN_IdentityD := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellD));

    v_LAI_D := cr_LAI(omit, v_PLMN_IdentityD, f_UTRAN_CellInfo_GetLAC(utran34_CellD));

    //Used for Equivalent PLMN List

    v_PLMN_IdentityE := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellE));

    // TMSI used throughout the test case is px_TMSI_Def;

    // new TMSI allocated will therefore always be the same

    v_U_LocationUpdatingRequest_Step4 :=  cr_U_LocationUpdatingRequestAll(?, // CKSN = initial value

                                                                          cr_LocUpdTypeNormal,

                                                                          -, // @sic R5s150829 Ch. 2.1 sic@

                                                                          cr_MS_Clsmk1_Def,

                                                                          -, // @sic R5s150829 Ch. 2.1 sic@

                                                                          cr_MS_Clsmk2_Any_tlv('33'O) ifpresent);

    v_U_LocationUpdatingRequest_Step15 :=  cr_U_LocationUpdatingRequestAll(?, // CKSN = initial value

                                                                           cr_LocUpdTypeNormal,

                                                                           v_LAI_A, // LAI = a

                                                                           cr_MS_Clsmk1_Def,

                                                                           cr_MobileIdTMSI_lv(px_TMSI_Def),// mobile identity = TMSI

                                                                           cr_MS_Clsmk2_Any_tlv('33'O) ifpresent);

    //@siclog "Step 1" siclog@

    //Set the cell type of cell B to the "Serving cell". Set the cell type of cell A to the "non-suitable cell".

    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellD, utran34_CellA); // @sic R5s150829 Ch. 2.2 sic@

    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationNonSuitableCell);

    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationNonSuitableCell);

    f_UTRAN_SetCellPower(utran34_CellC, tsc_AttenuationNonSuitableCell);

    f_UTRAN_SetCellPower(utran34_CellD, tsc_AttenuationSuitableNeighbourCell);

    //@siclog "Step 2" siclog@

    //The UE is switched on by either using the Power Switch or by applying power.

    //If PS Mode: The subsequent GMM attach should be accepted with "Equivalent PLMNs": PLMN E.

    //If PS mode: a routing area updating procedure should be performed.

    f_UT_SwitchOnUE(UT, false);

    f_Delay(v_Wait8minutes); //The SS waits a period of 8 minutes after the UE is switched on, this allowing the UE to make its first periodic search

    //@siclog "Step 3" siclog@

    //The SS verifies that the IE "Establishment cause" in the received RRC CONNECTION REQUEST message is set to "Registration"

    f_UTRAN34_RRC_ConnEst(utran34_CellD, registration);

    //If PS Mode: The subsequent GMM attach should be accepted with "Equivalent PLMNs": PLMN E.

    if (pc_PS == true) {

      t_WaitForGMM.start;

      interleave { //Receive in any order Location Area Update or GMM Attach

        //GMM attach

        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt,

                                                                                                                 tsc_I_AttachTypeCombined))))) -> value v_RRC_DataInd_PS

          {

          }

        //Location Area Update

        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step4)) -> value v_RRC_DataInd_CS

          {

          }

      }
      f_UTRAN_GMM_Authentication();

      f_UTRAN34_RRC_Security(utran34_CellD,                        // SECURITY MODE COMMAND, SECURITY MODE COMPLETE

                             true,

                             ps_domain);

      f_UTRAN34_AttachAccept(utran34_CellD, 

                             v_RRC_DataInd_PS.msg.attachRequest.attachType.attachType, 

                             -, -, -, -, -, -, -, -,

                             cs_PLMN_List_1PLMN(v_PLMN_IdentityE));       // ATTACH ACCEPT, ATTACH COMPLETE

      t_WaitForGMM.stop;

      //Send Location Update Accept

      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran34_CellD, v_RRC_DataInd_CS.ttcn_start, true, TMSI, cs_PLMN_List_1PLMN(v_PLMN_IdentityE));
    }

    else { // only CS

      //@siclog "Step 4 -8" siclog@

      //LOCATION UPDATING. "Location Update Type": Normal

      //The SS starts integrity protection. "Equivalent PLMNs": PLMN E

      f_UTRAN34_LocationUpdate(utran_CellDedicated, true, TMSI, v_U_LocationUpdatingRequest_Step4, -, cs_PLMN_List_1PLMN(v_PLMN_IdentityE));// @sic R5s150830 sic@

    }

    //@siclog "Step 9" siclog@

    //The SS releases the RRC connection

    f_UTRAN34_RRC_ConnRel(utran34_CellD, cell_Dch);

    //@siclog "Step 11" siclog@

    //Set the cell type of cell B to the "Suitable neighbour cell"

    //Set the cell type of cell C to the "Suitable neighbour cell"

    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationSuitableNeighbourCell);

    f_UTRAN_SetCellPower(utran34_CellC, tsc_AttenuationSuitableNeighbourCell);

    //@siclog "Step 12" siclog@

    //The SS shall wait for 8 minutes during which no messages should be received

    f_UTRAN_CheckNoDataInd(v_Wait8minutes, "Test Case 9.4.5.4.7, Step 12");

    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellA, utran34_CellD); // @sic R5s150829 Ch. 2.5 sic@

    //@siclog "Step 13" siclog@

    //Set the cell type of cell A to the "Suitable neighbour cell"

    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationSuitableNeighbourCell);

    //Within 8 minutes after step 13, the following messages shall be sent and received on Cell A

    //The lower boundary of time T shall not be checked in this test step

    t_Wait8minutes.start(v_Wait8minutes);

    //@siclog "Step 14" siclog@

    //The SS verifies that the IE "Establishment cause" in the received RRC CONNECTION REQUEST message is set to "Registration".

    U_TM.receive(car_RRC_ConnReq(utran34_CellA,

                                 tsc_RB0,

                                 cr_108_RRC_ConnReq_r8(registration))) -> value v_RLC_TR_DATA_IND

      { // @sic R5s150829 Ch. 2.5 sic@

        v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;

        f_UTRAN_CheckReleaseIndicator(v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator);

        f_UTRAN_MobileInfo_UE_Release_Set(v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator);

        f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);

      }

    if (pc_PS == true) {

      t_WaitForGMM.start;

      // wait for RAU and LU in any order

      interleave {

        //Receive Routing Area Update Request message

        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,

                                                  tsc_RB3,

                                                  cr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, v_UpdateType_Ra)))) -> value v_RRC_DataInd_RAU

          {

          }

        //Wait for Location Updating Request message

        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step15)) -> value v_RRC_DataInd_CS

          {

          }

      }

      t_WaitForGMM.stop;

      //SECURITY MODE COMMAND and SECURITY MODE COMPLETE

      f_UTRAN_GMM_Authentication();

      f_UTRAN34_RRC_Security(utran34_CellA, true, ps_domain);

      //Send ROUTING AREA UPDATE ACCEPT

      U_Dc.send(cas_DataReq_PS(utran_CellDedicated, tsc_RB3, cs_U_RA_UpdAcc3(cs_GMM_UpdateResult(v_UpdateType_Ra),

                                                                             cs_RAI(omit,

                                                                                    cs_LAI(omit, v_PLMN_IdentityA, f_UTRAN_CellInfo_GetLAC(utran34_CellA)),

                                                                                    f_UTRAN_CellInfo_GetRAC(utran34_CellA)),

                                                                             cs_PTMSI_Signature(px_PTMSI_SigDef),

                                                                             cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def),

                                                                             v_RRC_DataInd_RAU.msg.routingAreaUpdateRequest.pDP_ContextStatus)));

      //Receive ROUTING AREA UPDATE COMPLETE

      U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.4.7, Step 15");

      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran34_CellA, v_RRC_DataInd_CS.ttcn_start, true, TMSI);   // @sic R5s150829 Ch. 2.6 sic@
    }

    else { // only CS so just wait for LU

      //@siclog "Step 15-18" siclog@

      //LOCATION UPDATING REQUEST

      f_UTRAN34_LocationUpdate(utran_CellDedicated, true, TMSI, v_U_LocationUpdatingRequest_Step15); // @sic R5s150829 Ch. 2.6 sic@

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.4.7, Step 15");

    }

    t_Wait8minutes.stop;

    //@siclog "Step 19" siclog@

    //The SS releases the RRC connection

    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);

  }


4.4 TC_9_4_5_4_7()
	Record Name
	TC_9_4_5_4_7()

	Reason for change
	The current guardtimer for this testcase is set to 300s, which is insufficient for this testcase. It needs to be extended.

	Summary of change
	Increased to guard timer to 40*60s.

	Source of change
	UTRAN_Testsuite.ttcn

	MCC160 Comments
	Accepted


Before:

	testcase TC_9_4_5_4_7() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Location updating / periodic search for HPLMN or higher priority PLMN / UE waits Minimum Periodic search timer
    var UTRAN_PTC    v_UTRAN      := null;
    var GERAN_PTC    v_GERAN      := null;
    var EUTRA_PTC    v_EUTRA      := null;
    timer t_GuardTimer := int2float(300);  
    v_UTRAN := UTRAN_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_9_4_5_4_7_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }


After:

	testcase TC_9_4_5_4_7() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Location updating / periodic search for HPLMN or higher priority PLMN / UE waits Minimum Periodic search timer
    var UTRAN_PTC    v_UTRAN      := null;
    var GERAN_PTC    v_GERAN      := null;
    var EUTRA_PTC    v_EUTRA      := null;
    timer t_GuardTimer := int2float(40*60);//WA#9_4_5_4_7 300->2400
    v_UTRAN := UTRAN_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_9_4_5_4_7_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }


4.5 f_UTRAN34_AttachReject_LAU_REJ
	Record Name
	f_UTRAN34_AttachReject_LAU_REJ

	Reason for change
	Please see Change 4.2 ,Summary of Change Point 3 above.

	Summary of change
	New function

	Source of change
	UTRAN34_CommonFunctions.ttcn

	MCC160 Comments
	Accepted in principle. See implementation on following clause “Additional change from MCC160“


	function f_UTRAN34_AttachReject_LAU_REJ(UTRAN_CellId_Type  p_CellId,
                                  template (present) EstablishmentCause p_EstCause := registration) runs on UTRAN_PTC
  {
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var START_Value v_START_Value_cs := '00000000000000000000'B;
    var RRC_DATA_IND v_RRC_DataInd, v_RRC_DataInd_ps;
    var B2_Type v_NMO := f_UTRAN_CellInfo_GetNMO(p_CellId);
    var UE_OperationMode v_UE_OperationMode := f_UTRAN34_MobileInfo_UE_OperationMode_Get();
    var boolean v_GMM_Attach_received := false;
    var boolean v_LAU_received := false; 
    var RRC_DATA_IND v_RRC_DataInd_CS, v_RRC_DataInd_PS;
    //  Switch on UE
    f_UT_SwitchOnUE (UT, false);
    f_UTRAN34_RRC_ConnEst(p_CellId, p_EstCause); // Establish RRC connection
    interleave { //Receive in any order Location Area Update or GMM Attach
        //GMM attach
        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt,
                                                                                                                 tsc_I_AttachTypeCombined)),
                                                                                               cr_MobileIdAny(omit),
                                                                                               cr_RAI(omit),
                                                                                               ?))) -> value v_RRC_DataInd
          {
            v_GMM_Attach_received := true;
          }
        //Location Area Update
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_LocationUpdatingRequest)) -> value v_RRC_DataInd_CS
          {
            v_LAU_received := true;
          }
      }    
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                                   tsc_RB3,
                                   cs_U_AttachReject(tsc_RejCau_GPRS_SrvNotAllowed)));
    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_LocationUpdatingRej(tsc_RejCau_Illegal_ME)));
    f_UTRAN34_RRC_ConnRel(p_CellId, cell_Dch);
  }


4.6 Additional changes from MCC160

4.6.1 Function f_UTRAN34_AttachReject extended with parameter enum LOCATION_UPDATE_TYPE to include REJECT as an option to be performed, to reject the LAU
	  //---------------------------------------------------------------------

  /*

   * @desc      Generic test step to force UE to invalidate its SIM for PS parameters, i.e. to delete P-TMSI, P-TMSI signature, RAI and ciphering key sequence number.

   *            The test step shall allow either normal CS registration or reject the LAU; but always reject PS registration.

   *            Note-1: This generic test step shall correctly consider the different NMO, ATT flag and  UE mode of operation. (This is why the test step is so complex).

   *            Note-2: The UE is switched off when completing this test step

   * @param     p_CellId

   * @param     p_LOCATION_UPDATE_TYPE   (default value: TMSI) 

   *            Allowed values: REJECT - Reject the LAU Req

   *                            TMSI, IMSI, NONE - Accept the LAU Req with the appropriate value being included in the LAU Accept message

   * @param     p_EstCause          (default value: registration)

   * @status

   */
  function f_UTRAN34_AttachReject(UTRAN_CellId_Type  p_CellId,

                                  LOCATION_UPDATE_TYPE p_LOCATION_UPDATE_TYPE:= TMSI,

                                  template (present) EstablishmentCause p_EstCause := registration) runs on UTRAN_PTC

  {

    var RRC_DATA_IND v_RRC_DataInd_cs, v_RRC_DataInd_ps;

    var B2_Type v_NMO := f_UTRAN_CellInfo_GetNMO(p_CellId);

    var UE_OperationMode v_UE_OperationMode := f_UTRAN34_MobileInfo_UE_OperationMode_Get();

    //  Switch on UE

    f_UT_SwitchOnUE (UT, false);

    // UE is in Operation Mode A

    if (v_UE_OperationMode == opModeA) {

      // NMO I

      if (v_NMO == tsc_NMO_I or v_NMO == tsc_NMO_II_ExtendedNMO_I) {

        if (pc_AutomaticAttachSwitchON) {

          f_UTRAN34_AttachRejectProcedure(p_CellId, -, p_EstCause, tsc_I_AttachTypeCombined);

          //UE IMSI attach on CS
          f_UTRAN34_LocationUpdatingProcedure(p_CellId, -, p_EstCause, p_LOCATION_UPDATE_TYPE);

        } else {

          f_UTRAN34_LocationUpdatingProcedure(p_CellId, -, p_EstCause, p_LOCATION_UPDATE_TYPE);

          f_Delay (1.0);

          f_UT_TriggerAttach(UT);

          f_UTRAN34_AttachRejectProcedure(p_CellId, -, p_EstCause, tsc_I_AttachTypeCombined);

        }

      } else {

        // NMO II

        if (pc_AutomaticAttachSwitchON) {

          f_UTRAN34_RRC_ConnEst(p_CellId, p_EstCause); // Establish RRC connection
          // wait for RAU and LAU in any order and reject RAU

          interleave { // interleave does not allow calling user defined functions (TTCN standard V4.5.1 cl. 20.4)

            [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,

                                                      tsc_RB3,

                                                      cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeGPRSAtt)))) -> value v_RRC_DataInd_ps

              {}

            [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_LocationUpdatingRequest))-> value v_RRC_DataInd_cs

              {}

          }

          // Complete LAU first

          f_UTRAN34_LocationUpdatingProcedure(p_CellId, v_RRC_DataInd_cs, -, p_LOCATION_UPDATE_TYPE, noRrcConnectionRelease);

          // Reject ATTACH

          f_UTRAN34_AttachRejectProcedure(p_CellId, v_RRC_DataInd_ps);

        } else { // Not auto attach UE
          f_UTRAN34_LocationUpdatingProcedure(p_CellId, -, p_EstCause, p_LOCATION_UPDATE_TYPE);

          f_Delay (1.0);

          f_UT_TriggerAttach(UT);

          f_UTRAN34_AttachRejectProcedure(p_CellId, -, p_EstCause, (tsc_I_AttachTypeGPRSAtt, tsc_I_AttachTypeCombined));

        }

      }

    } else {

      // UE is in Operation Mode C - PS only

      if (not(pc_AutomaticAttachSwitchON)) {

        // UE does not automatically attempt PS attach

        // Trigger the Attach Req

        f_UT_TriggerAttach(UT);

      }

      f_UTRAN34_AttachRejectProcedure(p_CellId, -, p_EstCause, (tsc_I_AttachTypeGPRSAtt, tsc_I_AttachTypeGPRSAtt));

    }

  }


4.6.2 Optimization on interleave sections on related tests.
fl_TC_9_4_3_7_Body
	  function fl_TC_9_4_3_7_Body() runs on UTRAN_PTC

  {

    var RRC_DATA_IND v_RRC_DataInd_RAU;

    var NAS_Lac v_LAC_B;

    var UTRAN_CellInfo_Type v_CellInfoB := f_UTRAN_CellInfo_Get(utran34_CellB);

    var NAS_PlmnId v_PLMN_IdentityB;

    var template (present) LocAreaId v_LAI_B;

    var B1_Type v_DeviceProperties:= '1'B ;

    var float v_T3246 := 5.0;

    var float v_T3246_min := f_UG_SetTimerToleranceMin(v_T3246);

    var START_Value v_START_Value_PS := '00000000000000000000'B; //@sic R5s150674 change 4.4 sic@

    var START_Value v_START_Value_CS := '00000000000000000000'B;

    var RRC_DATA_IND v_RRC_DataInd_LAU;

    timer t_WaitForGMM := 5.0;

    // Variables to be used with cell B

    v_PLMN_IdentityB := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellB));

    v_LAC_B := v_CellInfoB.lac;

    v_LAI_B := cr_LAI(omit, v_PLMN_IdentityB, v_LAC_B);

    //@siclog "Step 1" siclog@

    //Set the cell type of cell B to the "Serving cell". Set the cell type of cell A to the "non-suitable cell".

    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationServingCell);

    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationNonSuitableCell);

    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA); // @sic R5s150674 change 4.4 sic@

    //@siclog "Step 2-4" siclog@

    //RRC CONNECTION Establishment

    //The SS verifies that the IE "Establishment cause" is set to "Delay Tolerant Access"

    f_UTRAN34_RRC_ConnEst(utran34_CellB, delayTolerantAccess);

    //If PS mode: a routing area updating procedure should be performed

    //@siclog "Step 5" siclog@

    //LOCATION UPDATING REQUEST. The SS verifies that the IE "Device properties" is set to "MS is configured for NAS signalling low priority"

    if (pc_PS == true) {

      t_WaitForGMM.start;

      // wait for RAU and LAU in any order

      interleave {

        //Receive Routing Area Update Request message

        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,

                                                  tsc_RB3,

                                                  cdr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, tsc_I_UpdateTypeRAOnly))))    //@sic R5s150682 sic@

          { }

        //Wait for Location Updating Request message

        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,

                                               tsc_RB3,

                                               cr_U_LocationUpdatingRequest_DeviceProperties(v_DeviceProperties)))

          {

            f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.3.7, Step 5");

          }

      }

      t_WaitForGMM.stop;

    }

    else { // only CS so just wait for LAU

      U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,

                                          tsc_RB3,

                                          cr_U_LocationUpdatingRequest_DeviceProperties(v_DeviceProperties)));

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.3.7, Step 5");

    }
    //@siclog "Step 6-7" siclog@

    //The SS waits for the disconnection of the main signalling link.

    //The SS includes the IE "Extended Wait Time" in the RRC CONNECTION RELEASE message. MS starts timer T3246 with the value of 5 seconds.

    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch, -, cs_RRCConnectionRelease_NonCriticalExtensions_ExtendedWaitTime(float2int(v_T3246)));

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.3.7, Step 6");

    //@siclog "Step 8" siclog@

    //The SS verifies that the UE does not initiate any communication before the T3246 timer has expired

    f_UTRAN34_RRC_RandAccFail(v_T3246_min);

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.3.7, Step 8");

    //@siclog "Step 9-11" siclog@

    //The SS verifies that the IE "Establishment cause" is set to "Delay Tolerant Access"

    f_UTRAN34_RRC_ConnEst(utran34_CellB, delayTolerantAccess);

    //@siclog "Step 12-14" siclog@

    //LOCATION UPDATING REQUEST

    //TMSI REALLOCATION COMPLETE

    if (pc_PS == true) {

        interleave { //@sic R5s150674 change 4.4 sic@

          //Receive Routing Area Update Request message

          [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,

                                                    tsc_RB3,

                                                    cdr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, tsc_I_UpdateTypeRAOnly)))) -> value v_RRC_DataInd_RAU   //@sic R5s150682 sic@

            {

              v_START_Value_PS := v_RRC_DataInd_RAU.ttcn_start;

            }

          //Wait for Location Updating Request message

          [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,

                                                 tsc_RB3,

                                                 cr_U_LocationUpdatingRequestAll(?, // CKSN = initial value

                                                                                 cr_LocUpdTypeNormal,

                                                                                 ?,

                                                                                 cr_MS_Clsmk1_Def,

                                                                                 f_Imsi2MobileIdentity(px_IMSI_Def),

                                                                                 cr_MS_Clsmk2_Any_tlv('33'O) ifpresent))) -> value v_RRC_DataInd_LAU

            {

              v_START_Value_CS := v_RRC_DataInd_LAU.ttcn_start;

            }

        }

        f_UTRAN34_LocationUpdate_WithoutLAUReq(utran34_CellB, v_START_Value_CS, true, TMSI);

        f_UTRAN34_GMM_RAU_Without_RAUReq(utran34_CellB,

                                         v_START_Value_PS,

                                         tsc_I_UpdateTypeRAOnly,

                                         omit,

                                         true,

                                         -,

                                         tsc_FOP_FollowOnProceed);

    }

    else {

       f_UTRAN34_LocationUpdate(utran34_CellB, true, TMSI);

    }
  }// fl_TC_9_4_3_7_Body end


fl_TC_9_4_5_5_Body
	    function fl_TC_9_4_5_5_Body() runs on UTRAN_PTC

  {

    var InitialUE_Identity v_InitialUE_Id;

    var U_RLC_TM_IND v_RLC_TR_DATA_IND;  //@sic R5-150124 sic@

    var RRC_DATA_IND v_RRC_DataInd_CS, v_RRC_DataInd_PS;

    var float v_T3212_min := (int2float(tsc_T3212)/1000.0) - 15.0; // 6 min - 15 seconds

    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step5;

    var template (present) RRCDataIndType v_U_LocationUpdatingRequest_Step13;

    var NAS_PlmnId v_PLMN_IdentityA;

    var template (present) LocAreaId v_LAI_A;

    var template (value) GPRS_Timer3 v_GprsTimer3 := cs_GPRS_Timer2_3_IEI('35'O, '101'B, '00110'B);

    timer t_WaitForAttach := 5.0;

    timer t_Wait := 30.0;

    // Variables to be used with cell A

    v_PLMN_IdentityA := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA));

    v_LAI_A := cr_LAI(omit, v_PLMN_IdentityA, f_UTRAN_CellInfo_GetLAC(utran34_CellA));

    // Templates to be used for Location Update messgages

    v_U_LocationUpdatingRequest_Step5 := cr_U_LocationUpdatingRequest_LocUpdType_MSNetworkFeatureSupport(?, // CKSN = initial value

                                                                                                         cdr_LocUpdTypeIMSI,

                                                                                                         v_LAI_A, // LAI = a

                                                                                                         cr_MobileIdTMSI_lv(px_TMSI_Def)); // mobile identity = TMSI

    v_U_LocationUpdatingRequest_Step13 :=  cr_U_LocationUpdatingRequestAll(?, // CKSN = initial value

                                                                           cdr_LocUpdTypePeriodic,

                                                                           v_LAI_A, // LAI = a

                                                                           cr_MS_Clsmk1_Def,

                                                                           cr_MobileIdTMSI_lv(px_TMSI_Def),

                                                                           cr_MS_Clsmk2_Any_tlv('33'O) ifpresent); // mobile identity = TMSI

    //@siclog "Step 1" siclog@

    //The UE is activated.

    f_UT_SwitchOnUE(UT, false);

    //@siclog "Step 2-4" siclog@

    //"Establishment cause": delayTolerantAccess @sic R5-153698 sic@

    f_UTRAN34_RRC_ConnEst(utran34_CellA, delayTolerantAccess); //@sic R5s150676 change 4.4 sic@

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.5, Step 2");

    //If PS mode: Any subsequent GMM attach from the UE shall be accepted by the SS.

    if (pc_PS == true) {

      t_WaitForAttach.start;

      interleave { //Receive in any order Location Area Update or Attach Request

        //GMM attach

        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt,

                                                                                                                 tsc_I_AttachTypeCombined))))) -> value v_RRC_DataInd_PS

          {}

        //Receive Location Area Update Request message

        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step5)) -> value v_RRC_DataInd_CS

          {

            f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.5, Step 5"); //@sic R5s150676 change 4.4 sic@

          }

      }

      //SECURITY MODE COMMAND and SECURITY MODE COMPLETE

      t_WaitForAttach.stop;

      f_UTRAN_GMM_Authentication();

      f_UTRAN34_RRC_Security(utran34_CellA, true, ps_domain);

      f_UTRAN34_AttachAccept(utran34_CellA, v_RRC_DataInd_PS.msg.attachRequest.attachType.attachType);// ATTACH ACCEPT, ATTACH COMPLETE

      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran34_CellA, v_RRC_DataInd_CS.ttcn_start, true, TMSI, omit, omit, v_GprsTimer3);  //@sic R5s150676 change 4.4 sic@

    }

    else { // only CS

      //@siclog "Step 5 -7" siclog@

      //LOCATION UPDATING REQUEST

      //"location updating type": IMSI attach

      //"MS network feature support": 1 (MS supports the extended periodic timer in this domain)

      f_UTRAN34_LocationUpdate(utran34_CellA, true, TMSI, v_U_LocationUpdatingRequest_Step5, -, omit, omit, v_GprsTimer3); // @sic R5s150830 sic@

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.5, Step 5");

    }
    //@siclog "Step 8-9" siclog@

    //SS waits for the disconnection of the main signalling link.

    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);

    //@siclog "Step 10" siclog@

    //"Establishment cause": delayTolerantAccess @sic R5-153698 sic@

    //This message shall be sent by the UE between 5 minutes 45 s and 6 minutes 15 s after step 9

    f_UTRAN34_RRC_RandAccFail(v_T3212_min);

    t_Wait.start;

    alt {

      [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,

                                      tsc_RB0,

                                      cr_108_RRC_ConnReq_r8(delayTolerantAccess))) -> value v_RLC_TR_DATA_IND

        {

          v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;

          f_UTRAN_CheckReleaseIndicator(v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.v3d0NonCriticalExtensions.v4b0NonCriticalExtensions.rrcConnectionRequest_v4b0ext.accessStratumReleaseIndicator);

        }

      [] t_Wait.timeout {

        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "RRC Connection Request not received within 5min45sec and 6min15sec after step 9");

      }

    }

    //@siclog "Step 11-12" siclog@

    f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);

    //@siclog "Step 13-14" siclog@

    //LOCATION UPDATING REQUEST

    f_UTRAN34_LocationUpdate(utran34_CellA, true, TMSI, v_U_LocationUpdatingRequest_Step13);

    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 9.4.5.5, Step 13"); //@sic R5-153698 sic@

    //@siclog "Step 15-16" siclog@

    //SS waits for the disconnection of the main signalling link.

    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);

  }

}


fl_TC_9_4_3_3a_Body
function fl_TC_9_4_3_3a_Body() runs on UTRAN_PTC

  {

…

    var B7_Type v_CauUnassNumber := '0000001'B;

 //   var boolean v_LAU_received := false;  // not needed any more

 //   var boolean v_RAU_received := false;  // not needed any more

 //   var boolean v_Attach_received := false; // not needed any more
…

   //@siclog "Step 5-6" siclog@

    //LOCATION UPDATING REQUEST. location updating type = normal, CKSN = initial value, LAI = b, mobile station classmark 1 as given by the ICS and mobile identity = TMSI.

    //LOCATION UPDATING REJECT. IE Reject cause is set to #22 in table 10.5.95 of TS 24.008, causes #2, #3, #6, #11, #12, #13 and #15 being excluded.

    if (pc_PS == true) {

      // wait for RAU and LAU in any order and reject

      t_WaitForGMM.start;

      interleave { // interleave does not allow calling user defined functions (TTCN standard V4.5.1 cl. 20.4)

        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,

                                                  tsc_RB3,

                                                  cdr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, ?)))) {} //@sic R5s150682 sic@

        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step5)) {}
      }

    }

    else { // only CS

      // wait for LAU and reject

      U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step5));

    }

    // if (pc_PS == true) RAU is rejected

    if (pc_PS == true) {
      t_WaitForGMM.stop;

      f_UTRAN34_SendRoutingAreaUpdateReject(tsc_RejCau_GPRS_SrvNotAllowed);

    }

    // if (pc_PS == true) or  (pc_CS == true) LAU is rejected    

    f_UTRAN34_SendLocationUpdatingReject(tsc_RejCau_Congestion); // @sic R5-144637 sic@ @sic R5s141390 sic@ @sic R5-150634 sic@

    //@siclog "Step 7-8" siclog@

    //The SS waits for the disconnection of the main signalling link.

    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);

…

    //@siclog "Step 82" siclog@

    //TMSI REALLOCATION COMPLETE

    if (pc_PS == true) {

      // wait for GMM Attach and LAU in any order and reject GMM Attach

      t_WaitForGMM.start;

      interleave { // interleave does not allow calling user defined functions (TTCN standard V4.5.1 cl. 20.4)

        [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt, tsc_I_AttachTypeCombined))))) -> value v_RRC_DataInd_ps {}
        [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, v_U_LocationUpdatingRequest_Step13))  -> value v_RRC_DataInd {}
      }

    }

    else { // only CS

      // wait for LAU

      f_UTRAN34_LocationUpdate(utran34_CellB, true, TMSI, v_U_LocationUpdatingRequest_Step13);

    }

    // if (pc_PS == true)  

    if (pc_PS == true) { 

      // 1. Attach Reject
      t_WaitForGMM.stop;

      v_START_Value_ps := v_RRC_DataInd_ps.ttcn_start;

      f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);

      U_Dc.send(cas_DataReq_PS(utran_CellDedicated,

                               tsc_RB3,

                               cs_U_AttachReject(tsc_RejCau_GPRS_SrvNotAllowed)));

      // 2. Handle LAU
      v_START_Value_ps := v_RRC_DataInd.ttcn_start;

      f_UTRAN34_LocationUpdate_WithoutLAUReq(utran34_CellB, v_START_Value_ps, true, TMSI);

    }

    //@siclog "Step 83-84" siclog@

5 Branches executed in test case 9.4.5.4.7
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 Intel 7360 device
The Intel 7260 device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_9_4_5_4_7_Intel_7360.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log provided, masking is applied to prevent any disclosure of confidential information.
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file

